Solving the continuous Sallen Key low pass equations using trapezoidal integration and equivalent currents



- **References**

http : // en.wikipedia.org / wiki / Capacitor
http : // en.wikipedia.org / wiki / Trapezoidal_rule
http : // en.wikipedia.org / wiki / Nodal_analysis
http : // qucs.sourceforge.net / tech / node26.html
http : // www.ecircuitcenter.com / SPICEtopics.htm

- **Trapezoidal integration of capacitor**

$$vc^{n+1} = vc^{n} + \frac{h}{C} \left(1/2\, ic^{n+1} + 1/2\, ic^{n}\right)$$

$$\frac{2\,C}{h} vc^{n+1} - \frac{2\,C}{h} vc^{n} = ic^{n+1} + ic^{n}$$

$$ic^{n+1} = \frac{2\,C}{h} vc^{n+1} - \frac{2\,C}{h} vc^{n} - ic^{n}$$

let

$$gc = \frac{2\,C}{h},\ iceq^{n} = gc\, vc^{n} + ic^{n}$$

then

$$\mathbf{ic^{n+1} = gc\, vc^{n+1} - iceq^{n}}$$

to update $iceq^{n+1}$ for the next time step after you have solved for vc^{n+1} and so ic^{n+1}

$$iceq^{n+1} = gc\, vc^{n+1} + ic^{n+1}$$

$$iceq^{n+1} = gc\, vc^{n+1} + gc\, vc^{n+1} - iceq^{n}$$

so

$$\mathbf{iceq^{n+1} = 2\, gc\, vc^{n+1} - iceq^{n}}$$

■ Solving for terms using nodal analysis (also called KCL, ie the sum of currents at each node is zero)

```
Clear["Global`*"];
nodev1 = g (v0 - v1) - (gc (v1 - k v2) - ic1eq) == 0
nodev2 = g (v1 - v2) - (gc (v2 - 0) - ic2eq) == 0
subst = {gc → 1}

Solve[{nodev1, nodev2} /. subst, {v1, v2}, {}][[1]] // FullSimplify
Solve[{nodev1, nodev2} /. subst, {v2}, {v0}][[1]] // FullSimplify
```

$$\text{ic1eq} + g\,(\text{v0} - \text{v1}) - \text{gc}\,(\text{v1} - k\,\text{v2}) == 0$$

$$\text{ic2eq} + g\,(\text{v1} - \text{v2}) - \text{gc}\,\text{v2} == 0$$

$$\{\text{gc} \to 1\}$$

$$\left\{\text{v1} \to \frac{\text{ic2eq}\,k + (1+g)\,(\text{ic1eq} + g\,\text{v0})}{(1+g)^2 - g\,k},\ \text{v2} \to \frac{\text{ic2eq} + g\,(\text{ic1eq} + \text{ic2eq} + g\,\text{v0})}{(1+g)^2 - g\,k}\right\}$$

$$\left\{\text{v2} \to \frac{\text{ic2eq} + g\,\text{v1}}{1+g}\right\}$$

■ Regrouping terms in bounded form

$$\text{FullSimplify}\left[\frac{\text{ic2eq}\,k + (1+g)\,(\text{ic1eq} + g\,\text{v0})}{(1+g)^2 - g\,k} - \left(\frac{k}{(1+g)^2 - g\,k}\,\text{ic2eq} + \frac{(1+g)}{(1+g)^2 - g\,k}\,\text{ic1eq} + \frac{(1+g)\,g}{(1+g)^2 - g\,k}\,\text{v0}\right)\right]$$

$$\text{FullSimplify}\left[\frac{\text{ic2eq} + g\,(\text{ic1eq} + \text{ic2eq} + g\,\text{v0})}{(1+g)^2 - g\,k} - \left(\frac{(1+g)}{(1+g)^2 - g\,k}\,\text{ic2eq} + \frac{g}{(1+g)^2 - g\,k}\,\text{ic1eq} + \frac{g^2}{(1+g)^2 - g\,k}\,\text{v0}\right)\right]$$

0

0

■ Plotting regrouped terms

```
a0 = 1 / ((1 + g)^2 - g k) /. {g → Tan[π wc]};
a1 = k a0 /. {g → Tan[π wc]};
a2 = (1 + g) a0 /. {g → Tan[π wc]};
a3 = g a2 /. {g → Tan[π wc]};
a4 = g a0 /. {g → Tan[π wc]};
a5 = g a4 /. {g → Tan[π wc]};

Show[
 Table[Plot[{a1, a2, a3, a4, a5}, {wc, 0, 1 / 2}, PlotRange → All], {k, 0, 2, 0.5}]]
```



Final algorithm

```
Clear
ic1eq = 0;
ic2eq = 0;

Set
g = Tan[π cutoff/samplerate];
k = 2*res;
a0 = 1/((1 + g)^2 - g k);
a1 = k*a0;
a2 = (1 + g)*a0;
a3 = g*a2;
a4 = g*a0;
a5 = g*a4;

Tick
(9*, 7+ , 16 total ops for low)

v1 = a1*ic2eq + a2*ic1eq + a3*v0;
v2 = a2*ic2eq + a4*ic1eq + a5*v0;
ic1eq = 2*(v1 - k*v2) - ic1eq;
ic2eq = 2*(v2       ) - ic2eq;

low = v2
```

Final algorithm using v1 to compute v2

```
Clear
ic1eq = 0;
ic2eq = 0;

Set
g = Tan[π cutoff/samplerate];
k = 2*res;
a0 = 1/((1 + g)^2 - g k);
a1 = k*a0;
a2 = (1 + g)*a0;
a3 = g*a2;
a4 = 1/(1 + g);
a5 = g*a4;

Tick
(8*, 6+ , 14 total ops for low)

v1 = a1*ic2eq + a2*ic1eq + a3*v0;
v2 = a4*ic2eq + a5*v1;
ic1eq = 2*(v1 - k*v2) - ic1eq;
ic2eq = 2*(v2       ) - ic2eq;

low = v2
```

Test of algorithm

```
ClearState[] := Block[{},
   ic1eq = 0;
   ic2eq = 0;
  ];

SetCoeff[cutoff_, res_, samplerate_] := Block[{},
   g = Tan[π cutoff / samplerate];
   k = 2 * res;
   a0 = 1 / ((1 + g)^2 - g k);
   a1 = k * a0;
   a2 = (1 + g) * a0;
   a3 = g * a2;
   a4 = g * a0;
   a5 = g * a4;
  ];

Tick[t_, v0_] := Block[{v1, v2, v3},
   v1 = a1 * ic2eq + a2 * ic1eq + a3 * v0;
   v2 = a2 * ic2eq + a4 * ic1eq + a5 * v0;
   ic1eq = 2 * (v1 - k * v2) - ic1eq;
   ic2eq = 2 * (v2) - ic2eq;
   Return[{t, v0, v2}]
  ];

h = 1.0 / 44 100.0;
t1 = 0.005;
drive = 1;
freq = 1000.0;

MySaw[x_] := 2 (x - Floor[x] - 0.5);
MyOsc[x_] := drive MySaw[freq x];

ClearState[];
SetCoeff[500.0, 0.5, 44 100.0];
tp0 = Table[Tick[t, MyOsc[t]], {t, 0, t1, h}];

ListPlot[{tp0[[All, {1, 2}]]}, PlotLabel → "input", Joined → True, PlotRange → All]
ListPlot[{tp0[[All, {1, 3}]]}, PlotLabel → "low", Joined → True, PlotRange → All]
```




low
0.2
0.1
−0.1
−0.2
−0.3
−0.4
0.001
0.002
0.003
0.004
0.005